D. Diode

Symbol Names: DIODE, ZENER, SCHOTTKY, VARACTOR, LED, TVS.

Syntax: Dnnn anode cathode <model> [area] [off] [m=<val>] [n=<val>] [temp=<value>]

Examples:

D1 SW OUT MyIdealDiode

.model MyIdealDiode D(Ron=.1 Roff=1Meg Vfwd=.4)

D2 SW OUT dio2

.model dio2 D(Is=1e-10)

Instance parameter M sets the number of parallel devices while instance parameter N sets the number of series devices.

A diode requires a .model card to specify its characteristics. There are two types of diodes available. One is a conduction region-wise linear model that yields a computationally light weight representation of an idealized diode. It has three linear regions of conduction: on, off and reverse breakdown. Forward conduction and reverse breakdown can non-linear by specifying a current limit with Ilimit(revIlimit). tanh() is used to fit the slope of the forward conduction to the limit current. The parameters epsilon and revepsilon can be specified to smoothly switch between the off and conducting states. A quadratic function is fit between the off and on state such that the diode's IV curve is continuous in value and slope and the transition occurs over a voltage specified by the value of epsilon for the off to forward conduction and revepsilon for the transition between off and reverse breakdown.

Below are the model parameters for this type of diode:

Name	Description	Units	Default
Ron	Resistance in forward conduction	Ω	1.
Roff	Resistance when off	Ω	1./Gmin
Vfwd	Forward threshold voltage to enter conduction	V	0.
Vrev	Reverse breakdown voltage	V	Infin.
Rrev	Breakdown impedance	Ω	Ron
Ilimit	Forward current limit	A	Infin.
Revilimit	Reverse current limit	A	Infin.
Epsilon	Width of quadratic region	V	0.
Revepsilon	Width of reverse quad. region	V	0.

This idealized model is used if any of Ron, Roff, Vfwd, Vrev or Rrev is specified in the model.

The other model available is the standard Berkeley SPICE semiconductor diode but extended to handle more detailed breakdown behavior and recombination current. The area factor determines the number of equivalent parallel devices of a specified model. Below are the diode model parameters for this diode.

Name	**Description**	**Units**	**Default**	**Example**
Is	saturation current	A	1e-14	1e-7
Rs	Ohmic resistance	Ω	0.	10.
N	Emission coefficient	-	1.	1.
Tt	Transit-time	sec	0.	2n
Cjo	Zero-bias junction cap.	F	0.	2p
Vj	Junction potential	V	1.	.6
M	Grading coefficient	-	0.5	0.5
Eg	Activation energy	eV	1.11	1.11 Si 0.69 Sbd 0.67 Ge
Xti	Sat.-current temp. exp	-	3.0	3.0 jn 2.0 Sbd
Kf	Flicker noise coeff.	-	0.	
Af	Flicker noise exponent	1.	1.	
Fc	Coeff. for forward-bias depletion capacitance formula	-	0.5	
BV	Reverse breakdown voltage	V	Infin.	40.
nbv	Reverse breakdown emission coefficient	-	1.	2.
Ibv	Current at breakdown voltage	A	1e-10	
Ibvl	Low-level reverse breakdown knee current	A	0.	
nbvl	Low-level reverse breakdown emission coefficient	-	1.	
Tnom	Parameter measurement temp.	°C	27	50
Isr	Recombination current parameter	A	0.	
Nr	Isr emission coeff.	-	2	
Ikf	High-injection knee current	A	Infin.	
Tikf	Linear Ikf temp coeff.	/°C	0.	
Trs1	linear Rs temp coeff.	/°C	0.	
Trs2	Quadratic Rs temp coeff.	/°C^2	0.	
Tbv1	Breakdown voltage temp coeff.	/°C	0.	
Tbv2	Quadratic breakdown voltage	/°C^2	0.	

	temp coeff.			
Tikf	Ikf temp coefficient	/°C	0.	
Perim	Default perimeter	m	0.	
Isw	Sidewall Is	A	0.	
ns	Sidewall emission coefficient	-	1.	
Rsw	Sidewall series resistance	Ω	0.	
Cjsw	Sidewall Cjo	F	0.	
Vjsw	Sidewall Vj	V	Vj	
mjsw	Sidewall mj	-	0.33	
Fcs	Sidewall Fc	-	Fc	

It is possible to annotate a model with voltage and current ratings. This information is displayed in the schematic capture GUI to assist in selecting a device but does not directly impact the electrical behavior in simulation. The following parameters may be specified.

Name	**Description**	**Units**
Vpk	Peak voltage rating	V
Ipk	Peak current rating	A
Iave	Ave current rating	A
Irms	RMS current rating	A
diss	Maximum power dissipation rating	W